
1 Overview
This document lists all the changes needed to correct issues in the ATS iWD-TTCN3-B2016-06_D17wk23 related to the title of this CR.

Contact:
Hans Rohnert

Hans.Rohnert.ext@rohde-schwarz.com
2 Corrections required

2.1 fl_IMS_Server_CheckPendingSecurityContext_ReleaseIPsec
	Function name
	fl_IMS_Server_CheckPendingSecurityContext_ReleaseIPsec

	Reason for change
	A run time error occurred during de-registration after Global Release is called. NW and UE address are not present in the global IMS server variable due to the initialization of the IMS Server.

	Summary of change
	Registration Address value is checked before sending IPSec Release after all TCP connections are closed.

	TTCN module
	Common/IP_PTC/IP_PTC_IMS_Handler.ttcn

	MCC160 Comment
	Accepted in principle:
there has been at least one other CR since the last delivery addressing the same area; furthermore MCC160 does not find an issue in the current TTCN with test case 8.1.1.3, as in f_IMS_DefaultHandler all TCP connections are closed already before f_IMS_GlobalRelease is called.
Nevertheless MCC160 can acknowledge an issue in case of f_IMS_GlobalRelease being called when there are still TCP connections, as f_IMS_GlobalRelease causes fl_IMS_Server_RemoveSecurityContext being called at the IP_PTC with p_WaitForCNF := false in which case it may come to the error as described when the pending security context shall be removed in a_IMS_TCP_Close_CNF_Handler.
NOTE: the proposed change in the CR would cause a pending security context not to be removed what may cause further side effects.


Before change
  function fl_IMS_Server_CheckPendingSecurityContext_ReleaseIPsec(inout IMS_Server_Type p_ImsServer,

                                                                  template (omit) IMS_SecurityContext_Type p_SecurityContext) runs on IP_PTC

  {

    var IMS_SecurityContext_Type v_PendingSecurityContext;

    var integer i;

    if (isvalue(p_SecurityContext)) {

      v_PendingSecurityContext := valueof(p_SecurityContext);

      if (ispresent(v_PendingSecurityContext.TcpClient) or ispresent(v_PendingSecurityContext.TcpServer)) {

        // => wait for confirmation of TCP close

        i := lengthof(p_ImsServer.PendingSecurityContextList);

        p_ImsServer.PendingSecurityContextList[i] := v_PendingSecurityContext;

      } else {

        // Release security association

        fl_IPsec_Release(p_ImsServer.RegistrationAddress.NW,

                         p_ImsServer.RegistrationAddress.UE,

                         v_PendingSecurityContext.SecurityInfo.Protected);

      }

    }

  }
After change
  function fl_IMS_Server_CheckPendingSecurityContext_ReleaseIPsec(inout IMS_Server_Type p_ImsServer,

                                                                  template (omit) IMS_SecurityContext_Type p_SecurityContext) runs on IP_PTC

  {

    var IMS_SecurityContext_Type v_PendingSecurityContext;

    var integer i;

    if (isvalue(p_SecurityContext)) {

      v_PendingSecurityContext := valueof(p_SecurityContext);

      if (ispresent(v_PendingSecurityContext.TcpClient) or ispresent(v_PendingSecurityContext.TcpServer)) {

        // => wait for confirmation of TCP close

        i := lengthof(p_ImsServer.PendingSecurityContextList);

        p_ImsServer.PendingSecurityContextList[i] := v_PendingSecurityContext;

      } else {

          if (isvalue(p_ImsServer.RegistrationAddress)) { 
          // Release security association

     
   fl_IPsec_Release(p_ImsServer.RegistrationAddress.NW,

                         p_ImsServer.RegistrationAddress.UE,

                         v_PendingSecurityContext.SecurityInfo.Protected);

          }
      }

    }

  }
MCC160 Implementation
	  type record IMS_PendingSecurityContext_Type {

    IMS_RegistrationAddress_Type        RegistrationAddress,

    IMS_SecurityContext_Type            SecurityContext

  };

	
  type record of IMS_PendingSecurityContext_Type IMS_PendingSecurityContextList_Type;

	  type record IMS_Server_Type {
    IMS_SecurityContext_Type            UnprotectedContext,

    IMS_SecurityContext_Type            ProtectedContext1   optional,

    IMS_SecurityContext_Type            ProtectedContext2   optional,

    IP_AddrInfo_Type                    IPv4Addr            optional,

    IP_AddrInfo_Type                    IPv6Addr            optional,

    IPsec_SecurityKeys_Type             SecurityKeys        optional,

    IMS_RegistrationAddress_Type        RegistrationAddress optional,

    PDN_Index_Type                      PdnIndex            optional,


    IMS_PendingSecurityContextList_Type PendingSecurityContextList
  };

	  template (value) IMS_PendingSecurityContext_Type cs_IMS_PendingSecurityContext(IMS_RegistrationAddress_Type p_RegistrationAddress,

                                                                                 IMS_SecurityContext_Type p_SecurityContext) :=

  { 
    RegistrationAddress := p_RegistrationAddress,

    SecurityContext := p_SecurityContext

  };

	  function fl_IMS_Server_CheckPendingSecurityContext_ReleaseIPsec(
                                                                  IMS_PendingSecurityContextList_Type p_PendingSecurityContextList,

                                                                  IMS_RegistrationAddress_Type p_RegistrationAddress,
                                                                  template (omit) IMS_SecurityContext_Type p_SecurityContext) runs on IP_PTC return IMS_PendingSecurityContextList_Type
  {
    var IMS_PendingSecurityContextList_Type v_PendingSecurityContextList := p_PendingSecurityContextList;
    var IMS_SecurityContext_Type v_SecurityContext;

    var integer i;

    if (isvalue(p_SecurityContext)) {

      v_SecurityContext := valueof(p_SecurityContext);

      if (ispresent(v_SecurityContext.TcpClient) or ispresent(v_SecurityContext.TcpServer)) {

        // => wait for confirmation of TCP close

        i := lengthof(v_PendingSecurityContextList);


        v_PendingSecurityContextList[i] := valueof(cs_IMS_PendingSecurityContext(p_RegistrationAddress, v_SecurityContext));
      } else {

        // Release security association

        fl_IPsec_Release(p_RegistrationAddress.NW,
                         p_RegistrationAddress.UE,

                         v_SecurityContext.SecurityInfo.Protected);

      }

    }
    return v_PendingSecurityContextList;
  }

	  function fl_IMS_Server_RemoveSecurityContext(inout IMS_Server_Type p_ImsServer,

                                               IMS_SecurityRelease_Type p_AllOrOldestOnly := oldestProtectedContextOnly,

                                               boolean p_WaitForCNF := true) runs on IP_PTC

  { /* @sic R5s141154: new parameter p_WaitForCNF sic@ */

    /* @sic R5-145732: support of 2nd security context sic@ */

    /* @sic R5s160735: p_AllOrOldestOnly sic@ */

    var IP_AddrInfo_Type v_NW_Address;

    var IP_AddrInfo_Type v_UE_Address;

    var IMS_SecurityInfo_Type v_SecurityInfo;

    var PortNumber_Type v_LocalPortServer;

    var PortNumber_Type v_LocalPortClient;

    var template (omit) IMS_SecurityContext_Type v_PendingSecurityContext;

    while (fl_IMS_Server_SecurityContext_IsStarted(p_ImsServer, protectedContext1)) { /* @sic R5s160735: "while" sic@ */

      v_NW_Address := p_ImsServer.RegistrationAddress.NW;

      v_UE_Address := p_ImsServer.RegistrationAddress.UE;

      v_SecurityInfo := p_ImsServer.ProtectedContext1.SecurityInfo.Protected;

      v_LocalPortServer := v_SecurityInfo.ProtectedPorts.Port_ps;

      v_LocalPortClient := v_SecurityInfo.ProtectedPorts.Port_pc;

      // stop UDP client

      f_UdpClient_Stop(v_NW_Address, v_LocalPortClient);

      // Close TCP connection

      v_PendingSecurityContext := fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, protectedContext1, p_WaitForCNF);

      if (not fl_IMS_Server_SecurityContext_IsStarted(p_ImsServer, protectedContext2)) {   // if there is no other security context ...

        // ... stop TCP and UDP server

        f_UdpServer_Stop(v_NW_Address, v_LocalPortServer);

        f_TcpServer_Stop(v_NW_Address, v_LocalPortServer, p_WaitForCNF);

        p_ImsServer.ProtectedContext1 := omit;

      }

      else {

        p_ImsServer.ProtectedContext1 := p_ImsServer.ProtectedContext2;

        p_ImsServer.ProtectedContext2 := omit;

      }


      p_ImsServer.PendingSecurityContextList := fl_IMS_Server_CheckPendingSecurityContext_ReleaseIPsec(p_ImsServer.PendingSecurityContextList, p_ImsServer.RegistrationAddress, v_PendingSecurityContext); // @sic R5s170203 sic@
      if (p_AllOrOldestOnly == oldestProtectedContextOnly) { /* @sic R5s160735 sic@ */

        break;

      }

    }

    if (p_AllOrOldestOnly == full) {  /* @sic R5s170231 - r1 of MCC160 comments: unprotected security context shall not be released in case of "oldestProtectedContextOnly" and no security context has been established yet sic@ */

      fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, unprotected, p_WaitForCNF); /* @sic R5s170231 - Additional changes: moved to here from a_IP_IMS_Config sic@ */

      p_ImsServer.RegistrationAddress := omit;  /* @sic R5s160825: p_ImsServer.RegistrationAddress needs to be reset as in case of IPv6 the UE may come up with a different interface ID at the registration sic@

                                                   @sic R5s170231 - Additional changes: moved to here from a_IP_IMS_Config sic@ */

    }

  }

	  altstep a_IMS_TCP_Close_CNF_Handler(inout IMS_Server_Type p_ImsServer) runs on IP_PTC

  {

    var IP_SOCKET_IND v_TCP_CLOSE_CNF;

    var IP_Connection_Type v_IP_Connection;

    var IMS_PendingSecurityContextList_Type v_SecurityContextList; 
    var IMS_PendingSecurityContextList_Type v_NewSecurityContextList; 
    var IMS_RegistrationAddress_Type v_RegistrationAddress;
    var IMS_SecurityContext_Type v_SecurityContext;

    var integer i;

    var integer v_Length;

    [] IP_SOCK_CTRL.receive(cr_TCP_CLOSE_CNF(?)) -> value v_TCP_CLOSE_CNF {

      v_SecurityContextList := p_ImsServer.PendingSecurityContextList;

      v_Length := lengthof(v_SecurityContextList);

      if (v_Length > 0) {
        v_NewSecurityContextList := {};

        v_IP_Connection := v_TCP_CLOSE_CNF.CTRL.ConnectionId;

        for (i := 0; i < v_Length; i := i + 1) {

          v_SecurityContext := v_SecurityContextList[i].SecurityContext; 
          v_RegistrationAddress := v_SecurityContextList[i].RegistrationAddress;
          if (match(v_SecurityContext.TcpClient, v_IP_Connection)) {

            v_SecurityContext.TcpClient := omit;

          } else if (match(v_SecurityContext.TcpServer, v_IP_Connection)) {

            v_SecurityContext.TcpServer := omit;

          }
          v_NewSecurityContextList := fl_IMS_Server_CheckPendingSecurityContext_ReleaseIPsec(v_NewSecurityContextList, v_RegistrationAddress, v_SecurityContext);

        }
        p_ImsServer.PendingSecurityContextList := v_NewSecurityContextList;
      }

    }

  }


Further Changes
	  function fl_IMS_Server_SetSecurityContext(inout IMS_Server_Type p_ImsServer,

                                            IMS_SecurityContextEnum_Type p_SecurityContextEnum,

                                            template (omit) IMS_SecurityContext_Type p_SecurityContext)

  {

    var IMS_SecurityContext_Type v_SecurityContext;

    if (isvalue(p_SecurityContext)) {

      v_SecurityContext := valueof(p_SecurityContext);

      select (p_SecurityContextEnum) {

        case (unprotected)       { p_ImsServer.UnprotectedContext := v_SecurityContext; }

        case (protectedContext1) { p_ImsServer.ProtectedContext1 := v_SecurityContext; }

        case (protectedContext2) { p_ImsServer.ProtectedContext2 := v_SecurityContext; }

      }

    } else {

      FatalError(__FILE__, __LINE__, "p_ImsServer expected to be present");     /* p_SecurityContext is declared as "template (omit)" just to avoid "valueof" at the calling side */    }

  }

	  function f_IMS_DefaultHandler() runs on IMS_PTC

  {

    var IMS_DefaultHandlerInfo_Type v_DefaultHandlerInfo;       /* @sic R5-163201 sic@ */

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var IMS_IPCAN_Coordination_MSG v_IMS_IPCAN_Coordination_MSG;

    var integer v_NonceCountForNextRegister;

    f_IMS_PTC_Init(-, defaultHandling, IMS_Security);

    v_DefaultHandlerInfo.State := WaitForRegistration;

    v_DefaultHandlerInfo.IpcanType := IPCAN_E;                  /* @sic R5s160133 sic@ */

    while (true) {

      alt {

        [] IPCAN.receive(cmr_IPCAN_IMS_Control) -> value v_IMS_IPCAN_Coordination_MSG {         /* @sic R5s160133 sic@ */

          v_DefaultHandlerInfo := fl_IMS_DefaultHandler_ChangeInfo(v_DefaultHandlerInfo, IPCAN_E, v_IMS_IPCAN_Coordination_MSG.IPCAN_IMS_Control);

        }

        [] OtherIPCAN.receive(cmr_IPCAN_IMS_Control) -> value v_IMS_IPCAN_Coordination_MSG {    /* @sic R5s160133 sic@ */

          v_DefaultHandlerInfo := fl_IMS_DefaultHandler_ChangeInfo(v_DefaultHandlerInfo, OtherIPCAN_E, v_IMS_IPCAN_Coordination_MSG.IPCAN_IMS_Control);

        }

        [] IMS_Server.receive(car_IMS_Register_Request(?)) -> value v_IMS_DATA_REQ {      // receive any REGISTER

          select (v_DefaultHandlerInfo.State) {

            case (WaitForRegistration) {

              f_IMS_DefaultHandler_InitialRegistration(v_DefaultHandlerInfo.IpcanType, v_IMS_DATA_REQ);  // @sic R5s140788 change 2 sic@

              v_DefaultHandlerInfo.State := WaitForDeregistration;

              v_NonceCountForNextRegister := 2;  /* next (re-)registration shall have nounce count of 2 @sic R5-163201 sic@ */

            }

            case (WaitForReregistration) {

              fl_IMS_DefaultHandler_Reregistration(v_DefaultHandlerInfo.IpcanType, v_IMS_DATA_REQ, v_NonceCountForNextRegister);  // @sic R5-163201 sic@

              v_NonceCountForNextRegister := v_NonceCountForNextRegister + 1;

            }

            case (WaitForDeregistration) {

              f_IMS_PTC_ResetRanType();                 // @sic R5s150057 change 3: reset RanType to allow de-registrtion from any RAT sic@

              f_IMS_PTC_Deregistration(v_IMS_DATA_REQ); // @sic R5-132037 sic@

              f_IMS_TCP_CloseHandling(waitAndClose);    /* @sic R5s170445: closing of TCP connection (after R5s140219 moving it into f_IMS_PTC_Deregistration) sic@ */ 
              f_IMS_GlobalRelease();
              v_DefaultHandlerInfo.State := WaitForRegistration;

            }

            case (NoRegistration) {                     // @sic R5-169150 sic@

              f_SetVerdictInconc(__FILE__, __LINE__, "IMS registration not allowed (e.g. as UE is in test loop mode)");

            }

            case else {

              FatalError(__FILE__, __LINE__, "IMS DefaultHandler: invalid state");

            }

          }

        }

      }

